Функции многих переменных, ряды

1. Дифференциальное исчисление функций многих переменных.
2. Локальные и условные экстремумы функции многих переменных.
3. Интегральное исчисление функций многих переменных.
4. Ряды и произведения (сумма ряда, произведение ряда, разложение функции в ряд).

Всюду далее примеры работы пакета Maple будут приведены в режиме интерфейса Worksheet Mode с текстовым режимом ввода команд и выражений (Text Mode). При выполнении заданий и упражнений используйте режим интерфейса Worksheet Mode с режимом ввода «двумерной математики» Maple Math Mode или ввода в строку Text Mode.

§1 Дифференциальное исчисление функций многих переменных

Большинство задач дифференциального и интегрального исчисления функций многих переменных решается в *Maple* теми же командами, что и для функций одной переменной, только с указанием дополнительных параметров.

Частные производные.

Для вычисления частных производных функции $f(x_1,\ldots,x_m)$ используется уже хорошо известная вам команда **diff**. В этом случае эта команда имеет такой формат: **diff(f,x1$n1,x2$n2,…, xm$nm)**, где **x1,…, xm** – переменные, по которым производится дифференцирование, а после знака **$** указаны соответствующие порядки дифференцирования. Например, частная производная $\frac{\partial^2 f}{\partial x \partial y}$ записывается в виде: **diff(f,x,y)**.

Для вычисления частных производных можно использовать шаблон $\frac{\partial}{\partial x} f$ на палитре Expressions (команда прямого исполнения **diff(f,x)**).

Задание 1.1.

1. Найти $\frac{\partial f}{\partial x}$ и $\frac{\partial f}{\partial y}$ функции $f = \operatorname{arctg} \frac{x}{y}$.

```
> f:=arctan(x/y):
> Diff(f,x)=simplify(diff(f,x));
```

$$\frac{\partial}{\partial x} \arctan \frac{x}{y} = \frac{y}{x^2 + y^2}$$

```
> Diff(f,y)=simplify(diff(f,y));
```

$$\frac{\partial}{\partial y} \arctan \frac{x}{y} = -\frac{x}{x^2 + y^2}.$$

2. Найти все частные производные 2-го порядка функции $f(x,y) = \frac{x-y}{x+y}$.

```
> restart; f:=(x-y)/(x+y):
> Diff(f,x$2)=simplify(diff(f,x$2));
```

$$\frac{\partial^2}{\partial x^2} \frac{x-y}{x+y} = -4\frac{y}{(x+y)^3}$$

```
> Diff(f,y$2)=simplify(diff(f,y$2));
```

$$\frac{\partial^2}{\partial y^2}\frac{x-y}{x+y}=4\frac{x}{(x+y)^3}$$

```
> Diff(f,x,y)=diff(f,x,y);
```

$$\frac{\partial^2}{\partial x \partial y}\frac{x-y}{x+y}=2\frac{x-y}{(x+y)^3}.$$

§2 Локальные и условные экстремумы функций многих переменных

Для исследования функции на локальный и условный экстремум используется команда из стандартной библиотеки **extrema(f,{cond},{x,y,…},'s')**, где **cond** – ограничения для поиска условного экстремума, которые записываются в виде равенств. После ограничений в фигурных скобках указываются все переменные, от которых зависит функция **f**, а затем в кавычках записывается **s** – имя переменной, которой будут присвоены координаты точек экстремума. Если ограничений не указывать, то будет производиться поиск локального экстремума.

К сожалению, команда **extrema** выдает все критические точки, то есть и те, в которых экстремума нет. Отсеять недающие экстремума критические точки можно с помощью непосредственной подстановки этих точек в функцию, например, оператором **subs**.

Так же, как и для функции одной переменной, наибольшее и наименьшее значения функции многих переменных вычисляются командами **maximize(f,{x1,…,xn},range)**, и **minimize(f,{x1,…,xn}, range)**, где следует указывать после функции в фигурных скобках список всех переменных, от которых она зависит, а затем интервалы для каждой переменной, указывающие область поиска наибольшего и наименьшего значений.

Если требуется найти переменные, при которых линейная функция многих переменных имеет максимум (или минимум) при выполнении определенных ограничений, заданных в виде линейных равенств или неравенств, то следует использовать симплекс-метод. Для этого сначала необходимо загрузит пакет **simplex**, а затем воспользоваться командой **maximize** (или **minimize**), где теперь в качестве **range** можно указывать в фигурных скобках ограничительную систему неравенств. Пакет **simplex** предназначен для решения задач линейной оптимизации. После его загрузки команды **maximize** и **minimize** меняют свое действие. Теперь эти команды выдают координаты точек, при которых заданная линейная функция имеет максимум или минимум. При этом допускается дополнительная опция для поиска только неотрицательных решений **NONNEGATIVE**.

Задание 2.1.

1. Найти экстремумы функции $f(x,y)=2x^4+y^4-x^2-2y^2$.

```
> restart:
> f:=2*x^4+y^4-x^2-2*y^2:
> extrema(f,{},{x,y},'s');s;
```

$$\{0, \frac{-9}{8}\}$$

$$\left\{\{x=0, y=-1\}, \{x=0, y=0\}, \{x=0, y=1\}, \left\{x=-\frac{1}{2}, y=-1\right\}, \left\{x=-\frac{1}{2}, y=0\right\}, \left\{x=-\frac{1}{2}, y=1\right\}, \left\{x=\frac{1}{2}, y=-1\right\}, \left\{x=\frac{1}{2}, y=0\right\}, \left\{x=\frac{1}{2}, y=1\right\}\right\}$$

Получилось всего два экстремума, поэтому очевидно, что f_{max}=0 и f_{min}=–9/8, причем максимум достигается в точке (0,0). Остальные критические точки следует проверить. В силу четности функции по обеим переменным, можно ограничиться проверкой критических точек только с положительными координатами.

```
> subs([x=1/2,y=1],f);
```

$$\frac{-9}{8}$$

```
> subs([x=1/2,y=0],f);
```

$$\frac{-1}{8}$$

```
> subs([x=0,y=1],f);
```

$$-1$$

Таким образом, функция имеет следующие локальные экстремумы: $f_{max}=f(0,0)=0$ и $f_{min}=f\left(\pm\frac{1}{2},\pm 1\right)=f\left(\pm\frac{1}{2},\mp 1\right)=-9/8$.

2. Найти наибольшее и наименьшее значения функции $f(x,y)=x^2+2xy-4x+8y$ в прямоугольнике x=0, y=0, x=1, y=2.

```
> restart:
> f:=x^2+2*x*y-4*x+8*y:
> maximize(f,x,y,x=0..1,y=0..2);
```

$$17$$

```
> minimize(f,x,y,x=0..1,y=0..2);
```

$$-3$$

Таким образом, функция имеет наибольшее значение $f_{max}=17$ и наименьшее значение $f_{min}=-3$.

3. Найти условные экстремумы функции $f(x,y,z)=xy+yz$ при $x^2+y^2=2$, $y+z=2$, $x>0$, $y>0$, $z>0$.

```
> restart: f:=x*y+y*z:
> assume(x>0);assume(y>0);assume(z>0);
> simplify(extrema(f,{x^2+y^2=2,y+z=2},{x,y,z},'s'));s;
```

$$\left\{2,\ -\frac{5}{2}-\frac{3}{2}\sqrt{3}\right\}$$

$$\left\{\{x\sim=-1, y\sim=1, z\sim=1\}, \{x\sim=1, y\sim=1, z\sim=1\}, \left\{x\sim=-\frac{1}{2}-\frac{1}{2}\sqrt{3}, y\sim=-\frac{1}{2}+\frac{1}{2}\sqrt{3}, z\sim=\frac{5}{2}-\frac{1}{2}\sqrt{3}\right\}, \left\{x\sim=-\frac{1}{2}+\frac{1}{2}\sqrt{3}, y\sim=-\frac{1}{2}-\frac{1}{2}\sqrt{3}, z\sim=\frac{5}{2}+\frac{1}{2}\sqrt{3}\right\}\right\}$$

Получили два экстремума и четыре экстремальных точки. В каких точках функция имеет экстремумы, можно определить подстановкой:

```
> subs(s[1],f);
```

$$0$$

```
> subs(s[2],f);
```

$$2$$

```
> simplify(subs(s[3],f));
```

$$-\frac{5}{2}+\frac{3}{2}\sqrt{3}$$

```
> simplify(subs(s[4],f));
```

$$-\frac{5}{2}-\frac{3}{2}\sqrt{3}$$

Таким образом, функция имеет следующие условные экстремумы: $f_{max}=f(1,1,1)=2$ и $f_{min}=f\left(-\frac{1}{2}+\frac{1}{2}\sqrt{3},\ -\frac{1}{2}-\frac{1}{2}\sqrt{3},\ \frac{5}{2}+\frac{1}{2}\sqrt{3}\right)=-\frac{5}{2}-\frac{3}{2}\sqrt{3}$

4. При каких значениях переменных функция $f(x,y,z)=-x+2y+3z$ имеет максимум, если требуется выполнение условий $x+2y-3z\le 4$, $5x-6y+7z\le 8$, $9x+10z\le 11$, а все переменные неотрицательные?

```
> restart: with(simplex):
> f:=-x+2*y+3*z:
> cond:={x+2*y-3*z<=4, 5*x-6*y+7*z<=8,
  9*x+10*z<=11}:
> maximize(f,cond,NONNEGATIVE );
```

$$\{x=0,\ y=\frac{73}{20},\ z=\frac{11}{10}\}$$

§3 Интегральное исчисление функций многих переменных

В *Maple* имеются две специальные команды для вычисления двойных и тройных интегралов, содержащиеся в библиотеке **student**.

Для вычисления двойных интегралов $\iint\limits_D f(x,y)dxdy$ используется команда **Doubleint(f(x, y), D)**, где **D** – область интегрирования, записываемая в одном из следующих форматов:

- **x=x1..x2, y=y1..y2**, где числа **x1, x2, y1, y2** задают прямоугольную область интегрирования;
- **x=f1(y)..f2(y), y=y1..y2**, где **f1(y), f2(y)** – линии, ограничивающие область интегрирования слева и справа на интервале от **y1** до **y2**;
- **x=x1..x2, y=g1(x)..g2(x)** , где **g1(y), g2(y)** – линии, ограничивающие область интегрирования снизу и сверху на интервале от **x1** до **x2**.

Для вычисления тройных интегралов $\iiint\limits_V f(x,y,z)dxdydz$ используется команда **Tripleint(f(x, y, z),x, y, z, V)**, где **V** – область интегрирования.

Обе эти команды являются командами отложенного действия. Чтобы получить значение интеграла, следует использовать команду **value(%)**.

Повторные интегралы можно вычислять с помощью вложения команды **int**, например, повторный интеграл $\int\limits_0^2 dy \int\limits_0^1 x^2y^3dx$ вычисляется командой

```
> int(int(x^2*y^3, x=0..1), y=0..2);
```

$$\frac{4}{3}$$

Задание 3.1.

1. Вычислить повторный интеграл $\int\limits_2^4 dy \int\limits_0^y \frac{y^3}{x^2+y^2}dx$

```
> Int(Int(y^3/(x^2+y^2),x=0..y),y=2..4)=
  int(int(y^3/(x^2+y^2), x=0..y),y=2..4);
```

$$\int\limits_2^4 dy \int\limits_0^y \frac{y^3}{x^2+y^2}dx = \frac{14}{3}\pi$$

2. Вычислить двойной интеграл $\iint\limits_D \sin(x+2y)dxdy$ по области, ограниченной линиями $y=0,\ y=x,\ x+y=\frac{\pi}{2}$.

Замечание: сначала следует описать область интегрирования D в виде неравенств: $D=\{(x,y): y \le x \le \frac{\pi}{2}-y,\ 0 \le y \le \frac{\pi}{2}\}$

```
> restart: with(student):
> J:=Doubleint(sin(x+2*y), x=y..Pi/2-y, y=0..Pi/2);
```

$$J := \int_0^{\frac{1}{2}\pi} \int_y^{\frac{1}{2}\pi-y} \sin(x+2y)dxdy$$

```
> J:=value(%);
```

$$J := \frac{2}{3}$$

3. Вычислить тройной интеграл $\int\limits_{-1}^{1} dx \int\limits_{x^2}^{1} dy \int\limits_{0}^{2} (4+z)dz$.

Замечание: следует помнить, что порядок интегрирования определяется последовательностью пределов, поэтому сначала внутренние указываются пределы, содержащие функции.

```
> J:=Tripleint(4+z, y=x^2..1,x=-1..1, z=0..2);
```

$$J := \int_0^2 \int_{-1}^1 \int_{x^2}^1 4+zdydxdz$$

```
> J:=value(%);
```

$$J := \frac{40}{3}$$

§4. Ряды и произведения

Вычисление суммы ряда и произведений

Конечные и бесконечные суммы $\sum\limits_{n=a}^{b} S(n)$ вычисляются командой прямого исполнения **sum** и отложенного исполнения **Sum**. Аргументы этих команд одинаковые: **sum(expr, n=a..b)**, где **expr** – выражение, зависящее от индекса суммирования, **a..b** – пределы индекса суммирования, указывающие, что суммировать следует от **n=a** до **n=b**.
Команду прямого исполнения **sum(f, i=k..n)** можно набрать с помощью шаблона  на палитре Expressions.

Если требуется вычислить сумму бесконечного ряда, то в качестве верхнего предела вводится **infinity**.

Аналогичным образом вычисляются произведения $\prod\limits_{n=a}^{b} P(n)$ командами прямого **product(P(n),n=a..b)** и отложенного действий **Product P(n),n=a..b).**

Команду прямого исполнения **product(f, i=k..n)** можно набрать с помощью шаблона $\prod_{i=k}^{n} f$ на палитре Expressions.

Задание 4.1.

1. Найти полную и N-частичную суммы ряда, общий член которого равен: $a_n = \frac{1}{(3n-2)(3n+1)}$.

```
> restart: a[n]:=1/((3*n-2)*(3*n+1));
> S[N]:=Sum(a[n], n=1..N)=sum(a[n], n=1..N);
> S:=limit(rhs(S[N]), N=+infinity);
> S:=sum(a[n], n=1..+infinity);
```

$$a_n := \frac{1}{(3\,n-2)\,(3\,n+1)}$$

$$S_N := \sum_{n=1}^{N} \frac{1}{(3\,n-2)\,(3\,n+1)} = -\frac{1}{3\,(3\,N+1)} + \frac{1}{3}$$

$$S := \frac{1}{3}$$

$$S := \frac{1}{3}$$

2. К какой функции сходится степенной ряд: $\sum_{n=1}^{\infty} (-1)^{n+1} n^2 x^n$?

```
> Sum((-1)^(n+1)*n^2*x^n, n=1..infinity)=
  sum((-1)^(n+1)*n^2*x^n, n=1..infinity);
```

$$\sum_{n=1}^{\infty} (-1)^{n+1}\, n^2\, x^n = -\frac{x\,(x-1)}{(x+1)^3}$$

3. Найти сумму степенного ряда $\sum_{n=0}^{\infty} \frac{(1+x)^n}{(n+1)n!}$.

```
> Sum((1+x)^n/((n+1)*n!), n=0..infinity)=
  sum((1+x)^n/((n+1)*n!), n=0..infinity);
```

$$\sum_{n=0}^{\infty} \frac{(x+1)^n}{(n+1)\,n!} = \frac{e^{x+1}-1}{x+1}$$

4. Вычислить бесконечное произведение: $\prod_{n=2}^{\infty} \frac{n^3-1}{n^3+1}$

```
> Product((n^3-1)/(n^3+1),n=2..infinity)=
  product((n^3-1)/(n^3+1), n=2..infinity);
```

$$\prod_{n=2}^{\infty} \frac{n^3-1}{n^3+1} = \frac{2}{3}.$$

<u>Разложение функции в степенной ряд и ряд Тейлора.</u>

Разложение функции $f(x)$ в степенной ряд в окрестности точки x_0

$$f(x)=\sum_{n=1}^{\infty} a_n(x-x_0)^n = a_0 + a_1(x-x_0) + a_2(x-x_0)^2 + \ldots + a_n(x-x_0)^n + \ldots$$

осуществляется командой **series(f(x), x=x0, n)**, где **x0** – точка, в окрестности которой производится разложение, **n** – порядок разложения.

Аналогичного действия команда **taylor(f(x), x=a, n)** раскладывает функции **f(x)** в окрестности точки **x=a** по формуле Тейлора:

$$f(a)+\frac{f'(a)}{1!}(x-a)+\frac{f''(a)}{2!}(x-a)^2+\ldots+\frac{f^{(n)}(a)}{n!}(x-a)^n+\ldots$$

Команды **series** и **taylor** выдают результат, имеющий тип series. Для того, чтобы иметь возможность дальнейшей работы с полученным разложением, его следует преобразовать в полином с помощью команды **convert(%,polynom)**.

Функцию многих переменных $f(x_1,\ldots,x_n)$ можно разложить в ряд Тейлора по набору переменных $(x_1,\ldots,x_n)$ в окрестности точки $(a_1,\ldots,a_n)$ до порядка n с помощью команды **mtaylor(f(x), [x1,…,xn], n)**.

Задание 4.2.

1. Разложить в степенной ряд $f(x)=e^{-x}\sqrt{x+1}$ в окрестности x_0=0, удерживая 5 первых членов.

```
> f(x)=series(exp(-x)*sqrt(x+1), x=0, 5);
```

$$f(x)=1-\frac{1}{2}x-\frac{1}{8}x^2+\frac{13}{48}x^3-\frac{79}{384}x^4+O(x^5)$$

2. Построить на одном рисунке графики интеграла ошибок $\operatorname{erf}(x)=\frac{2}{\sqrt{\pi}}\int_0^x e^{-t^2}dt$ и его разложения в ряд Тейлора в окрестности нуля.

```
> taylor(erf(x),x,8): p:=convert(%,polynom);
```

$$p:=2\frac{x}{\sqrt{\pi}}-\frac{2}{3}\frac{x^3}{\sqrt{\pi}}+\frac{1}{5}\frac{x^5}{\sqrt{\pi}}-\frac{1}{21}\frac{x^7}{\sqrt{\pi}}$$

```
> plot([erf(x),p],x=-2..2,thickness=[2,2],
  linestyle=[1,3], color=[red,green]);
```



Пунктирной линей изображен график ряда Тейлора, а сплошной – самой функции.

3.Разложить функцию двух переменных $f(x,y)=\sin(x^2+y^2)$ в ряд Тейлора в окрестности точки (0, 0) до 6-ого порядка.

```
> f=mtaylor(sin(x^2+y^2), [x=0,y=0], 7);
```

$$f=x^2+y^2-\frac{1}{6}x^6-\frac{1}{2}y^2x^4-\frac{1}{2}y^4x^2-\frac{1}{6}y^6$$